

Mise au point Software pour Cartouches Super-Oric (12)

Cercles Concentriques de toutes les couleurs!

par André C.

Lors de la dernière Visu, Fabrice F. nous a, une fois de plus, épatés avec un petit programme Basic tapé 'au vol' sur le Super-Oric, grâce au clavier de Thierry B. Le présent article vous propose ce listing et quelques variantes. Peu de personnes disposent d'une console Snes, d'une cartouche Snes modifiée (pour recevoir une mémoire flash à la place de la Rom d'origine) et encore moins d'un clavier. Toutefois, il est possible pour tout un chacun de bricoler des programmes pour le Super-Oric, en utilisant quelques outils de base dont un émulateur Snes. J'ai déjà développé cette procédure dans cette série d'articles et ne vais pas revenir dessus. Si vous êtes un adhérent récent, il est possible que vous ne disposiez pas de ces anciens articles. Je me ferai une joie de vous les envoyer sur simple demande.

UN PEU DE BIBLIOGRAPHIE

Beaucoup d'articles ont été consacrés au Super-Oric dans le Ceo-Mag. Je limiterai ma liste à la série consacrée à la programmation, dont fait partie le présent article.

- n°173, p7-8: Mise en cartouche d'un programme Basic
- n°174, p14-16: Programmation du Joypad
- n°176, p12-15: L'écran Hires du Super-Oric
- n°177, p14-17: Notions sur la structure de l'écran Hires
- n°178, p14-17: L'écran Texte
- n°179, p13-15: 32768 Couleurs!
- n°180, p9-12: 32768 Couleurs! (suite et fin)
- n°181, p9-12: Caractères Redéfinis (1)
- n°182, p8-10: Caractères Redéfinis (2)
- n°183-184, p9-14: Caractères Redéfinis (3)
- n°188, p8-12: L'affichage en vidéo inverse

LE LISTING DE FABRICE À LA VISU

```
10 FOR I=0TO100
20 DEF INK I, I
30 NEXT
40 HIRES
50 CURSET 128,120,3
100 FOR I=1TO100
110 INK I
120 CIRCLE I, 1
130 NEXT
```

et le résultat (ci-contre): Que voit-on? Des cercles concentriques de rayon I croissant de 1 à 100, tracés dans la couleur du premier plan (si nécessaire, révisez votre Basic en consultant votre manuel Atmos). Cette couleur varie pour chaque cercle tracé. En effet les lignes 10 à 30 définissent les couleurs n°0 à 100 et indique leur définition avec un paramètre variant lui aussi de 0 à 100 (pour la sim-

plicité de la programmation). Rappelons que la couleur n°0 est celle du papier et est donc définie ici par la valeur 0 (noir). Question: pourquoi la couleur des cercles ne varie pas régulièrement, mais forme trois zones (une petite 4e est inachevée), variant de manière identique du noir (en fait rouge très foncé) au rouge vif? Réponse: parce que, dans le Super-Oric, les rouges purs sont codés sur 5 bits, ce qui fait 31 valeurs possibles. Seule la zone centrale est tracée avec des rouges purs, les deux autres contiennent un soupçon de vert, mais cela ne se voit pas. Le résultat est assez joli, non?

RAPPEL SUR LE CODAGE DES COULEURS


J'ai traité ce sujet en profondeur dans les mags n°179 & 180 (32768 Couleurs!). Voici, pour la clarté de la suite, quelques explications.

Les couleurs sont codées sur 2 octets: 5 bits pour chaque composante (r=rouge, v=vert & b=bleu) et un bits inutilisé qui reste toujours à zéro:

`0bbb bbvv vvvr rrrr`

Composante Bleue

Le bleu le plus vif est défini ainsi: 0111 1100 0000 0000 (en binaire), ce qui se traduit directement par #7C00 en hexadécimal et donne 31744 en décimal. En fait, les 31 bleus purs sont codés du plus sombre au plus vif de 0000 0100 0000 0000 (soit #0400 ou 1024 en décimal) à 0111 1100 0000 0000 (soit #7C00 ou 31744). Les 5 bits codent bien de 1 à 31 pour la composante bleue proprement dite et il faut multiplier cette valeur par 1024 pour refléter la position des 5 bits dans les deux octets de définition globales. On vérifie que cela donne bien des valeurs de $1 \times 1024 = 1024$ à $31 \times 1024 = 31744$.



Composante verte

Il en va de même pour les 31 verts, dont la composante proprement dite est également codée sur 5 bits et dont la position dans la définition globale est reflétée par un facteur de multiplication $\times 32$. Les 31 verts purs sont donc codés du plus sombre au plus vif de 0000 0000 0010 0000 (soit #0020 et 32 en décimal) [et on a bien $1 \times 32 = 32$] à 0000 0011 1110 0000 (soit #03E0 et 992 en décimal) [et on a bien $31 \times 32 = 992$].

Composante rouge

Pour les rouges, c'est encore plus simple, car le facteur multiplicatif est $\times 1$. Les rouges purs sont donc simplement codés de 1 à 31 soit de: 0000 0000 0000 0001 (ou #0001) à 0000 0000 0001 1111 (ou #001F). Revenons à la figure obtenue par Fabrice, les 31 premiers cercles vont du rouge sombre au rouge vif. On a ensuite un cercle vert très sombre pour la valeur: 0000 0000 0010 0000 (#20 ou 32). Mais il faudrait une loupe pour le voir! Puis la couleur des 31 cercles suivants est 'enrichie' d'un peu de vert foncé, grâce à ce 'petit' bit vert qui passe à 1 pour les couleurs jusqu'à la valeur 63. Ensuite, 64 s'écrit 0000 0000 0100 0000 (#40) et donc rebelote on a un cercle vert sombre suivit de 31 cercles rouges 'enrichis' d'un soupçon de vert foncé. L'œil n'est pas capable de faire la différence entre les trois zones 'rouges'.

Le tracé de la figure de Fabrice n'est pas terminé. Son programme définit 101 couleurs (noir+100 rouges ou 'presque rouges') et trace 100 cercles. Nous en étions à $31+1+31+1+31=95$, il reste 5 cercles, eux aussi dans une variante de 'rouge'. On les voit à la périphérie et cette fois on distingue nettement la composante verte qui devient non négligeable.


VARIANTES

Nous n'avons pas fait tout ce travail de compréhension pour le seul plaisir de comprendre (ce qui n'est déjà pas si mal), mais aussi pour être capables de jouer avec les couleurs du Super-Oric. Je ne sais pas si vous avez remarqué, mais les cercles de Fabrice sont légèrement décentrés sur la gauche. C'est parce qu'il a tapé sa démo vite fait sur le gaz! Or l'écran Hires comporte 256 pixels horizontalement (axe des x de 0 à 255 de gauche à droite) et 224 pixels verticalement (axe des y de 0 à 223 de haut en bas). Il n'y a donc pas de centre exact, mais on peut prendre $x=128$ et $y=112$, la différence restera invisible.

Il est aussi possible de tracer plus de cercles, mais il ne faut pas sortir de l'écran, sinon on obtient une "?ILLEGAL QUANTITY ERROR". On est donc limité à 111 cercles (en bas $112+111=223$ et en haut $112-111=1$). Si on modifie le listing de Fabrice avec:

```
10 FOR I=0TO111
100 FOR I=1TO111
```

on obtient la figure suivante (haut page suivante): On remarque que la 4e zone est un peu plus étendue et passe nettement dans les 'rouges verdaces'.




ON POUSSE LES VERTS!

Juste pour rigoler, on ne va plus prendre des couleurs de valeurs 'continues' mais tracer une couleur sur huit. Pour ce faire on remplace le début du programme par:

```
10 FOR I=0TO111
15 J=I*8
20 DEF INK I,J
30 NEXT
```


Le résultat est le suivant:



Certes, cela ne nous surprend pas, mais l'esthétique de la chose est douteuse et plutôt que de faire n'importe quoi au hasard, essayons d'être rationnels. Essayons par exemple de faire une figure analogue à celle de Fabrice, mais dans les verts. Au vu de nos considérations sur le codage des verts, le programme devient:

```
10 FOR I=0TO111
15 J=I*32
20 DEF INK I,J
30 NEXT
```

Et le résultat comble nos espoirs (page suivante): En examinant la figure, on constate que la 4e zone donne cette fois dans les bleus.



CERCLES ROUGES, VERTS ET BLEUS


Nous commençons à maîtriser. De plus en plus fort, essayons de tracer trois zones: une rouge, une verte et une bleue. On va donc commencer à limiter le nombre de cercles à $31 \times 3 = 93$. En plus de la couleur 0 pour le papier, on va définir trois séries de 31 couleurs correspondant aux rouges purs, aux verts purs et aux bleus purs. Cela donne le programme suivant:

```

10 FOR I=0 TO 31
20 DEF INK I, I
30 NEXT
40 FOR I=1 TO 31
50 DEF INK I+31, I*32
60 NEXT
70 FOR I=1 TO 31
80 DEF INK I+62, I*1024
90 NEXT
100 HIRES
110 CURSET 128, 112, 3
120 FOR I=1 TO 93
130 INK I
140 CIRCLE I, 1
150 NEXT

```

Programme que nous lançons avec impatience... Et



bien non, ça ne donne pas un affreux coloriage, mais bel et bien ce que nous voulions (en bas à gauche): On voit tout de suite que chacune des trois couleurs évolue d'une teinte très foncée, proche du noir, vers une teinte très vive. Et il n'y a plus de 4e zone.

Y a-t-il un moyen de condenser les trois boucles de définitions des couleurs ? Imaginons par exemple:

```


10 DEF INK 0, 0
20 FOR I=1 TO 31
30 DEF INK I, I
40 DEF INK I+31, I*32
50 DEF INK I+62, I*1024
60 NEXT

```

On ne gagne que trois lignes, mais c'est plus joli. Inconvénient: les couleurs ne sont pas définies linéairement de 0 à 93, mais 0, 1, 32, 63, 2, 33, 64 etc. Le Super-Oric va-t-il y retrouver ses petits? Et bien oui, comme le montre la figure suivante (en bas à droite), parfaitement identique à la précédente. C'est un enseignement intéressant: on peut définir et redéfinir les couleurs de la palette dans n'importe quel ordre. Rappel : Dans chaque ligne DEF INK, le premier paramètre est le n° de la première couleur à définir, le (ou les) paramètre(s) suivant(s) correspond(ent) à la définition de ce caractère (et éventuellement des caractères de numéros consécutifs qui suivent le premier numéro indiqué).

CERCLES JAUNES, MAGENTAS ET CYANS

Et si pour varier un peu on faisait la même chose avec les couleurs jaunes, magentas et cyans? Vous devez bien vous douter que ça va se compliquer, puisque le jaune est un mélange de rouge et de vert sans bleu: le magenta est un mélange de rouge et vert sans bleu, le cyan un mélange de vert et bleu sans rouge. Et tout ça peut se mélanger dans les proportions les plus variables. Il n'y a plus 31 nuances, mais $31 \times 31 = 961$ possibilités. Pour réduire le problème, nous nous en tiendrons aux couleurs jaunes, magentas et cyans pures, où la proportion des deux composantes est la même. Ce qui réduit notre problème à 31 nuances pour chaque zone. Ouf! Enfin presque, car il va falloir cogiter un peu quand même...



Pour les jaunes, on mélange du rouge et du vert soit: DEF INK I,(I)+(I*32) soit DEF INK I,I*33 Vous suivez, oui? on ajoute du rouge (I) et du vert (I*32) voir le listing précédent. Les couleurs de la première zone, n° 1 à 31, qui correspondent donc au jaunes purs, seront définies par les valeurs 33 à 1023.


Pour les magentas, on mélange du rouge et du bleu soit: DEF INK I+31,(I)+(I*1024) ou DEF INK I,I*1025 Les couleurs n° 32 à 62, qui correspondent donc au magentas purs, seront définies par les valeurs 1025 à 31775.

Pour les cyans, on mélange du vert et du bleu: DEF INK I+62,(I*32)+(I*1024) ou DEF INK I+62,I*1056 Les couleurs n° 63 à 93, qui correspondent donc au cyans purs, seront définies par les valeurs 1056 à 32736.

Au final, cela donne le programme suivant:

```
10 DEF INK 0,0
20 FOR I=1 TO 31
30 DEF INK I , I*33
40 DEF INK I+31, I*1025
50 DEF INK I+62, I*1056
60 NEXT
100 HIRES
110 CURSET 128,112,3
120 FOR I=1 TO 93
130 INK I
140 CIRCLE I,1
150 NEXT
```

Et voici ce qu'on obtient:




32768 COULEURS?

Vous me direz: "Bon tout ça, ce ne sont jamais que des couleurs Oric plus ou moins sombres et pour les autres couleurs?". Et bien oui, nous n'avons fait qu'effleurer les possibilités du Super-Oric. Depuis le début, nous n'avons affiché qu'un peu moins de 500 couleurs et il en reste plus de 32000! Un abîme sans fond! Je laisse à votre expérimentation les 31 nuances de gris (mélange à égalité de rouge, vert et bleu) à titre d'exercice. Ce qui ne devrait pas être bien dif-

ficile en modifiant un peu le programme précédent. Allons faisons-nous plaisir en affichant quelques beaux oranges. Les oranges sont formés en théorie par un mélange de rouge et de jaune. Nous venons de voir que le jaune est lui-même un mélange de rouge et de vert. Pour faire des oranges (sic) il faut donc un mélange de rouge (en proportion principale) et de vert. En fait, certains oranges contiennent aussi un peu de bleu (voir exemples dans le mag n°180). Nous allons voir ce que donne la figure suivante: une première zone avec un maximum de rouge, une gamme de vert et pas de bleu du tout, une 2e zone semblable mais avec un apport constant d'un soupçon de bleu et enfin une 3e zone avec un soupçon plus net de bleu. Soit le listing:

```
10 DEF INK 0,0
20 FOR I=1 TO 31
30 DEF INK I , 31+I*32
40 DEF INK I+31, 31+I*32+10*1024
50 DEF INK I+62, 31+I*32+16*1024
60 NEXT
100 HIRES
110 CURSET 128,112,3
120 FOR I=1 TO 93
130 INK I
140 CIRCLE I,1
150 NEXT
```

Et voici le résultat: on voit immédiatement que seule la première zone (maximum de rouge et gamme de vert) contient des oranges. Cette zone va du rouge au jaune (celui-ci étant obtenu avec autant de vert que de rouge). Les deux autres zones sont légèrement violacée (le 2e) ou carrément mauve (la 3e). Dans les deux cas, l'apport croissant de vert finit par l'emporter et les derniers cercles sont à nouveau jaune, quoiqu'un peu sale.



Note: Fabrice avait également tapé un deuxième listing lors de la visu (voir les photos dans le mag de juillet) mais je l'ai mal recopié et n'ai pas réussi à reproduire la figure qu'il avait obtenue.